

Programmation Fonctionnelle L2 – Examen Partiel 17/3/2022

1 Types

Considérer le programme suivant.

```

1 f1 x = x++"abc"
2 f2 x y = x:y:"bcd"
3 f3 x y = x (x y)
4 f4 x y z = if x then y else z
5 f5 x y = x:f5 x (tail y)
6 f6 x = map x x
7 f7 x y = [ z<y | z<-x ]
8 f8 x y = filter (x y)

```

- Trouver les lignes qui donnent une erreur de compilation. Expliquer quelle est l'erreur. Pour les expressions f_1, f_2, \dots, f_8 pour lesquelles le programme n'a pas d'erreur de compilation, donner le type inféré par le compilateur. (4 points)

Solution :

f_6 a une erreur de compilation, car le premier argument de `map` doit être une fonction et le deuxième une liste, donc `x` ne peut pas être les deux à la fois.

```

1 f1 :: [Char] -> [Char]
2 f2 :: Char -> Char -> [Char]
3 f3 :: (t -> t) -> t -> t
4 f4 :: Bool -> p -> p -> p
5 f5 :: t -> [a] -> [t]
6 f7 :: Ord a => [a] -> a -> [Bool]
7 f8 :: (t -> a -> Bool) -> t -> [a] -> [a]

```

Barème : 0,5 points par fonction (pour le bon type ou pour détection d'erreur de type)

2 Types – Part 2

Pour chacun des types suivants, donner une expression **correcte** de Haskell qui a ce type, c'est-à-dire, le compilateur infère le type donné pour votre expression. (2 points)

```

1 Char -> [Bool]
2 Eq a => a -> Bool
3 [a] -> (a->b) -> [b]
4 (a -> b) -> [a] -> [(a,b)]

```

Solution :

```

1 g1 x = if x=='a' then [True] else [False]
2 g2 x = x==x
3 g3 xs f = map f xs
4 g4 f xs = zip xs (map f xs)

```

Barème : 0,5 points par type. Évidemment il y a d'autres solutions possibles, vous pouvez les vérifier avec `ghci`.

3 Nombres de Hamming

Un entier positif x est un **nombre de Hamming** si $x = 2^a 3^b 5^c$, où a, b, c sont des entiers positifs. Par exemple $24 = 2^3 3^1 5^0$ et $60 = 2^2 3^2 5^1$ sont des nombres de Hamming, alors que 14 n'est pas un nombre de Hamming.

Écrire une fonction `firstHamming` qui prend comme argument un entier n et renvoie les n premiers nombres de Hamming.

Exemple d'utilisation:

```
*Main> firstHamming 25
[1,2,3,4,5,6,8,9,10,12,15,16,18,20,24,25,27,30,32,36,40,45,48,50,54]
```

(3 points)

Solution :

```
1 merge :: Ord a => [a] -> [a] -> [a]
2 merge [] xs = xs
3 merge xs [] = xs
4 merge (x:xs) (y:ys)
5   | x<y = x:merge xs (y:ys)
6   | x==y = merge xs (y:ys)
7   | otherwise = y:merge (x:xs) ys
8
9 merge3 :: Ord a => [a] -> [a] -> [a] -> [a]
10 merge3 a b c = merge (merge a b) c
11
12 hamming :: [Integer]
13 hamming = 1:merge3 (map (2*) hamming) (map (3*) hamming) (map (5*) hamming)
14
15 firstHamming :: Int -> [Integer]
16 firstHamming n = take n hamming
17
18 isHamming :: Integer -> Bool
19 isHamming 1 = True
20 isHamming x
21   | x`mod`2==0 = isHamming (x`div`2)
22   | x`mod`3==0 = isHamming (x`div`3)
23   | x`mod`5==0 = isHamming (x`div`5)
24   | otherwise = False
25 -- compare performance of
26 -- filter isHamming [1..10^7]
27 -- takeWhile (<=10^7) hamming
```

4 Flip

La fonction `flip` de Haskell a comme type `flip :: (a -> b -> c) -> b -> a -> c`. Elle prend comme argument une fonction `f` de deux arguments et inverse le rôle de ces deux arguments. Par exemple :

```
*Main> flip (:) [1,2,3] 4
[4,1,2,3]
*Main> flip (-) 5 3
-2
```

Donner une implémentation de la fonction `flip`. (2 points)

Solution :

```
1 flip f x y = f y x
```

5 Map, Fold, etc.

Quels sont les résultats d'évaluation des lignes suivantes ? Est-ce que toutes les lignes sont correctes ou pouvez vous trouver des erreurs de compilation ? (5 points)

NB: la fonction `flip` est celle de l'exercice 4.

```
1 foldl (flip (:)) "abc" "def"
2 foldr ($) 5 [(+3), (*2), (9-)]
3 map (length.tail) ["hello", "my", "name", "is", "Haskell"]
4 [ x y y | x<-[(+), (*), (-)], y<-[1..3] ]
5 zipWith (\x y -> if y==0 then Nothing else Just $ x`div`y) [3..7] [-2..2]
```

Solution :

```
foldl (flip (:)) "abc" "def"
"fedabc"
```

```
foldr ($) 5 [(+3), (*2), (9-)]
11
```

```
map (length.tail) ["hello", "my", "name", "is", "Haskell"]
[4,1,3,1,6]
```

```
[ x y y | x<-[(+), (*), (-)], y<-[1..3] ]
[2,4,6,1,4,9,0,0,0]
```

```
zipWith (\x y -> if y==0 then Nothing else Just $ x`div`y) [3..7] [-1..3]
[Just (-2), Just (-4), Nothing, Just 6, Just 3]
```

Barème : 1 point par question

6 Matchings

Supposons qu'on a deux listes $xs :: [a]$ et $ys :: [b]$ **de la même taille**. Un couplage (matching) de ces deux listes est une liste zs de type $[(a,b)]$ telle que chaque élément x de xs et chaque élément y de ys paraît exactement une fois dans un tuple de zs .

Exemple : Disons que trois amis, John, Paul, et George, veulent former un groupe musical¹. On a donc $xs = ["John", "Paul", "George"]$.

Leurs instruments sont

$ys = ["Guitar", "Piano", "Drums"]$.

Un couplage entre ces deux listes est une affectation qui nous dit quel membre du groupe va jouer quel instrument. Par exemple

$zs1 = [("John", "Piano"), ("Paul", "Guitar"), ("George", "Drums")]$

est un couplage valide. Par contre,

$zs2 = [("John", "Piano"), ("Paul", "Guitar"), ("John", "Drums")]$

n'est pas valide, car ça donne deux instruments à John et rien à George.

Pour les questions suivantes vous pouvez supposer que les listes xs, ys ont la même taille et ne contiennent pas de doublons.

1. Écrire une fonction `isMatching` qui prend comme arguments les listes xs, ys, zs et retourne `True` si zs est un couplage valide de xs, ys .
2. Écrire une fonction `allMatchings` qui prend comme arguments les listes xs, ys et retourne une liste zss qui contient tous les couplage valides de xs, ys (dans n'importe quel ordre).
3. Écrire une fonction `allMatchingsConstrained` qui a le même comportement que `Matchings` sauf qu'elle prend un troisième argument cs . La liste cs contient des affectations interdites. Par exemple, si John ne veut pas jouer la guitare et Paul ne veut pas jouer la batterie on mettrait

$cs = [("John", "Guitar"), ("Paul", "Drums")]$

Votre fonction doit retourner une liste de tous les couplages qui évitent les affectations interdites par cs .

Exemples d'utilisation :

```
*Main> isMatching [1,2,3] "abc" (zip [3,2,1] "cab")
```

True

```
*Main> isMatching [1,2,3] "abc" (zip [3,2,2] "abc")
```

False

```
*Main> allMatchings [1,2] "ab"
```

```
[[ (1,'a'), (2,'b') ], [ (1,'b'), (2,'a') ]]
```

```
*Main> allMatchings [1,2,3] "abc"
```

```
[[ (1,'a'), (2,'b'), (3,'c') ], [ (1,'a'), (2,'c'), (3,'b') ], [ (1,'b'), (2,'a'), (3,'c') ],
 [ (1,'b'), (2,'c'), (3,'a') ], [ (1,'c'), (2,'a'), (3,'b') ], [ (1,'c'), (2,'b'), (3,'a') ]]
```

```
*Main> allMatchingsConstrained [1,2,3] "abc" [(1,'a'), (1,'c')]
```

```
[[ (1,'b'), (2,'a'), (3,'c') ], [ (1,'b'), (2,'c'), (3,'a') ]]
```

```
*Main> allMatchingsConstrained [1,2,3] "abc" [(1,'a'), (2,'b')]
```

```
[[ (1,'b'), (2,'a'), (3,'c') ], [ (1,'b'), (2,'c'), (3,'a') ], [ (1,'c'), (2,'a'), (3,'b') ]]
```

(5 points)

Solution :

```
1 -- First solution
2
3 allMatchings :: [a] -> [b] -> [ [(a,b) ] ]
4 allMatchings [] _ = [[]]
```

¹Ringo va les rejoindre plus tard. ;-)

```

5 allMatchings _ [] = [[]]
6 allMatchings (x:xs) ys = concat [map ((x,ys!!i):) (allMatchings xs (remove i ys))
7                               | i<-[0..length ys-1]
8
9 remove :: Int -> [a] -> [a]
10 remove 0 xs = tail xs
11 remove i xs = head xs:remove (i-1) (tail xs)
12
13 -- Second solution
14
15 matchings' :: [a] -> [b] -> [[(a,b)]]
16 matchings' [] _ = [[]]
17 matchings' _ [] = [[]]
18 matchings' xs ys = map (zip xs) (allperms ys)
19
20 allperms :: [a] -> [[a]]
21 allperms [x] = [[x]]
22 allperms (x:xs) = concatMap (puteverywhere x) (allperms xs)
23
24 puteverywhere :: a -> [a] -> [[a]]
25 puteverywhere x [] = [[x]]
26 puteverywhere x (y:ys) = (x:y:ys): map (y:) (puteverywhere x ys)
27
28 -- Matching with Constraints
29
30 allMatchingsConstrained :: (Eq a, Eq b) => [a] -> [b] -> [(a,b)] -> [ [(a,b)] ]
31 allMatchingsConstrained xs ys [] = allMatchings xs ys
32 allMatchingsConstrained xs ys ((x,y):zs) =
33     filter (not. elem (x,y)) $ allMatchingsConstrained xs ys zs
34
35 -- Attn: below we rely on the assumption that xs, ys have no duplicates
36
37 isMatching :: (Eq a, Eq b) => [a] -> [b] -> [(a,b)] -> Bool
38 isMatching xs ys zs = (length xs == length zs)
39                       && subset xs (map fst zs)
40                       && subset ys (map snd zs)
41
42 subset :: Eq a => [a] -> [a] -> Bool
43 subset [] _ = True
44 subset (x:xs) ys = elem x ys && subset xs ys

```

Annexe

Consignes

- Essayer de justifier vos réponses aussi bien que possible.
- Pour toutes les fonctions que vous définissez, vous devez préciser une signature de type.
- Pour les exercices de programmation, vous avez le droit de définir d'autres fonctions secondaires en plus des fonctions demandées (mais il faut aussi bien préciser leurs types).
- Vous pouvez utiliser toutes les fonctions de Haskell de la liste suivante. (Pour chaque fonction on vous rappelle aussi son type). Toute autre fonction que vous utilisez doit être implémentée.

Fonctions de base

```
(:) :: a -> [a] -> [a]
head :: [a] -> a
tail :: [a] -> [a]
last :: [a] -> a
(++) :: [a] -> [a] -> [a]
(!!) :: [a] -> Int -> a
length :: [a] -> Int
take :: Int -> [a] -> [a]
drop :: Int -> [a] -> [a]
reverse :: [a] -> [a]
repeat :: a -> [a]
(.) :: (b -> c) -> (a -> b) -> a -> c
($) :: (a -> b) -> a -> b
map :: (a -> b) -> [a] -> [b]
zip :: [a] -> [b] -> [(a, b)]
filter :: (a -> Bool) -> [a] -> [a]
concatMap :: (a -> [b]) -> [a] -> [b]
foldl :: (b -> a -> b) -> b -> [a] -> b
foldr :: (a -> b -> b) -> b -> [a] -> b
dropWhile :: (a -> Bool) -> [a] -> [a]
zipWith :: (a -> b -> c) -> [a] -> [b] -> [c]
fst :: (a, b) -> a
snd :: (a, b) -> b
(^) :: (Num a, Integral b) => a -> b -> a
(+) :: Num a => a -> a -> a
(-) :: Num a => a -> a -> a
(*) :: Num a => a -> a -> a
(==) :: Eq a => a -> a -> Bool
(/=) :: Eq a => a -> a -> Bool
(/) :: Fractional a => a -> a -> a
div :: Integral a => a -> a -> a
mod :: Integral a => a -> a -> a
max :: Ord a => a -> a -> a
min :: Ord a => a -> a -> a
(<) :: Ord a => a -> a -> Bool
(<=) :: Ord a => a -> a -> Bool
id :: a -> a
```